

Лабораторная работа №2

ОСНОВЫ ЯЗЫКА GPSS

Цель работы – изучение основных элементов языка моделирования GPSS

1. Теоретическая часть

1.1. Формат операторов языка GPSS

Операторы в GPSS бывают двух видов – блоки и команды. Блоки соответствуют статическим объектам модели. Команды управляют ходом моделирования и описывают некоторые используемые объекты программы.

Операторы в GPSS имеют следующий формат (в квадратных скобках указаны необязательные элементы):

[<Метка (имя)>] <Операция> <Операнды> [;<Комментарий>]

Метка представляет собой последовательность символов, начинающуюся с буквы, и используется для пересылки транзактов к данному блоку.

Операция представляет собой название блока или команды, основные из которых подробно рассматриваются далее. В GPSS принято записывать названия блоков заглавными буквами.

Блоки могут иметь операнды. Состав и значение операндов зависят от типа блока. Операнды принято обозначать буквами A, B, C, D, E, F, G. Часть операндов может быть необязательными. Операнды определяются по номеру позиции. Поэтому, например, если операнд A опущен, а операнд B присутствует, то нужно записать “ , 1 ”, показывая запятой, что значение “1” относится ко второму операнду.

Операндами могут быть не только константы, но и выражения. В выражениях могут присутствовать константы, СЧА, вызовы функций и знаки операций. В GPSS используются следующие знаки операций: “+” – сложение, “-” – вычитание, “#” – умножение, “/” – деление, “\” – целочисленное деление, “@” – остаток от деления. При использовании внутри операнда вызова функции с параметрами рекомендуется весь операнд заключать в круглые скобки.

Значением некоторых операндов может быть имя блока. Имя блока представляет собой алфавитно-цифровую последовательность длиной до 250 символов, которая начинается с буквы. Допускается использование латинских букв, цифр и символа подчеркивания.

Одной из полезных возможностей GPSS является команда EQU, позволяющая использовать в программе именованные константы. Она имеет формат:

<Имя константы> EQU <значение>

Константа должна быть объявлена до момента ее первого использования.

1.2. Ввод транзактов в модель. Блок GENERATE

Блок GENERATE предназначен для ввода транзактов в модель. Он имеет следующий формат:

GENERATE [A],[B],[C],[D]

В квадратных скобках показаны необязательные параметры. Блок имеет 5 параметров, каждый из которых может отсутствовать. Значения параметров приведены в следующей таблице:

Параметр	Описание	Значение по умолчанию
A	Средний интервал времени между поступлениями двух соседних транзактов	0
B	Половина величины интервала для равномерного распределения	0
C	Время генерации первого транзакта	0
D	Максимальное число генерируемых транзактов	бесконечность

Несколько примеров использования данного блока приведены в таблице:

Генерирование транзактов через фиксированные интервалы в 5 единиц времени	GENERATE 5
Генерирование транзактов через равномерно распределенный от 5 до 7 интервал времени (6 ± 1)	GENERATE 6,1
Генерирование транзактов через 3 единицы времени, начиная с момента времени 20	GENERATE 3,,20
Генерирование одного транзакта	GENERATE,,,1

1.3. Вывод транзактов из модели. Блок TERMINATE

Блок TERMINATE предназначен для вывода транзактов из модели. Данный блок имеет один параметр, который задает, на сколько уменьшается **счетчик завершения**. Счетчик завершения – это системная переменная, начальное значение которой устанавливается командой START. После этого при каждом вхождении транзакта в блок TERMINATE счетчик завершения уменьшается на величину, заданную в параметре A (по умолчанию – не изменяется). Когда счетчик завершения достигнет 0, моделирование останавливается.

1.4. Моделирование работы систем в течение заданного времени

В рассмотренном нами в первой работе примере моделирование останавливалось по условию обслуживания определенного числа клиентов. Часто встает другая задача – исследование работы системы в течение заданного периода времени. В этом случае модель меняют таким образом, чтобы счетчик завершения отсчитывал не обслуженных клиентов, а время работы. Модель из первой работы для этого нужно модифицировать следующим образом. Последний блок “TERMINATE 1” заменить на “TERMINATE 0” для того, чтобы транзакт, моделирующий обслуженного клиента, удалялся из модели, не изменяя счетчик завершения. После данного блока нужно добавить следующие два:

```
GENERATE 1
TERMINATE 1
```

Таким образом, каждую единицу времени будет создаваться и сразу уничтожаться один транзакт, уменьшая при этом счетчик завершения. Период времени, в течение которого выполняется моделирование, задается в команде START.

1.5. Моделирование временных задержек транзактов. Блок ADVANCE

Для моделирования временных задержек в GPSS предназначен блок ADVANCE. После входа в данный блок транзакт задерживается в нем на некоторое время, и только после этого выходит из блока. Формат блока:

```
ADVANCE A[,B]
```

Параметр А задает среднее время задержки, а параметр В – величину половины интервала для равномерного закона распределения.

Примеры использования данного блока приведены в таблице:

Задержка на 5 единиц времени	ADVANCE 5
Задержка на 10 ± 5 единиц времени	ADVANCE 10,5

1.6. Моделирование очередей. Блоки QUEUE, DEPART

Для сбора статистики о вынужденном ожидании транзактов в модели предназначены блоки QUEUE и DEPART. Блок QUEUE является входом в очередь, а блок DEPART – выходом из очереди. Время, которое транзакт движется от блока QUEUE к соответствующему блоку DEPART, считается временем нахождения транзакта в очереди. Данные блоки имеют следующий формат:

QUEUE A[,B]
DEPART A[,B]

Параметр А задает имя очереди, а параметр В – число единиц, на которое увеличивается (в блоке QUEUE) или уменьшается (в блоке DEPART) длина очереди.

Статистика по очередям выдается в специальном окне QUEUE ENTITIES, которое доступно через команду меню “Window/Simulation window/Queues window”. Основные параметры устройств описаны в таблице:

Название параметра	Описание
Queue Entity	Имя очереди
Current Content	Текущая длина очереди
Maximum Content	Максимальная длина очереди
Average Content	Средняя длина очереди
Average Time (+0)	Среднее время ожидания (с учетом транзактов, время ожидания которых равно 0)
Average Time (-0)	Среднее время ожидания (без учета транзактов, время ожидания которых равно 0)

1.7. Моделирование устройств. Блоки SEIZE, RELEASE

Для моделирования устройств в GPSS предназначена пара блоков – SEIZE и RELEASE. Блок SEIZE соответствует входу в устрой-

ство, а блок RELEASE – выходу из него. Оба блока имеют один параметр – имя устройства. После того, как один транзакт вошел в блок SEIZE, все устройство считается занятым, и другие транзакты не могут в него войти до тех пор, пока первый не выйдет из соответствующего блока RELEASE, тем самым, освободив устройство.

Блоки SEIZE и RELEASE не обязательно должны использоваться попарно. Допустимой является ситуация, когда одному устройству в модели соответствуют несколько блоков SEIZE или RELEASE. При входе транзакта в один из блоков SEIZE все остальные также становятся занятыми. А при выходе транзакта из одного из блоков RELEASE все входы освобождаются.

Статистика по устройствам выдается в специальном окне FACILITY ENTITIES, которое доступно через команду меню “Window/Simulation window/Facilities window”. Основные параметры устройств описаны в таблице:

Название параметра	Описание
Facility	Имя устройства
Utilization	Коэффициент использования
Ave. Time	Среднее время пребывания транзакта в устройстве

1.8. Моделирование многоканальных устройств. Блоки ENTER, LEAVE

Многоканальное устройство может одновременно обслуживать несколько транзактов. В начале программы каждое многоканальное устройство должно быть описано с помощью директивы STORAGE. Она имеет следующий формат:

<Имя> STORAGE A

Где <Имя> – имя многоканального устройства, а параметр A задает число каналов обслуживания.

Далее в модели многоканальное устройство реализуется блоками ENTER и LEAVE. Блок ENTER является входом в многоканальное устройство, а блок LEAVE – выходом из него. Данные блоки имеют следующий формат:

ENTER A[,B]
LEAVE A[,B]

Параметр А задает имя многоканального устройства, а параметр В – количество занимаемых транзактом каналов обслуживания.

Так же как и для обычных устройств, в модели может быть несколько блоков ENTER или LEAVE для одного многоканального устройства.

Статистика по многоканальным устройствам выдается в специальном окне STORAGE ENTITIES, которое доступно через команду меню “Window/Simulation window/Storages window”. Основные параметры многоканальных устройств описаны в таблице:

Название параметра	Описание
Storage	Имя многоканального устройства
Utilization	Коэффициент использования
Capacity	Число каналов обслуживания
Storage In Use	Число занятых каналов (на текущий момент)
Min In Use	Минимальное число занятых каналов
Max In Use	Максимальное число занятых каналов

1.9. Блок TRANSFER

Для того чтобы транзакт переходил в какой-либо блок, отличный от последующего, в GPSS используется блок TRANSFER. При этом блок для перехода указывается меткой, которая также записывается перед названием блока. Блок TRANSFER может работать в одном из нескольких режимов. Рассмотрим два из них.

Режим безусловной передачи. Блок имеет следующий формат:

TRANSFER ,B

Параметр В задает метку блока, на который осуществляется переход. Например, для того, чтобы передать транзакт на блок TERMINATE, используется следующая конструкция:

TRANSFER ,EXIT

...

EXIT TERMINATE

Статистический режим. В данном режиме блок передается на один из двух блоков с некоторой вероятностью. Формат блока:

TRANSFER A,[B],C

Значения параметров описаны в таблице:

Параметр	Описание	Значение по умолчанию
A	Вероятность передачи транзакта в блок C, задаваемая в долях тысячи	Ошибка
B	Позиция блока, в которую должен перейти транзакт с вероятностью 1-A	Следующий по порядку блок
C	Позиция блока, в которую должен перейти транзакт с вероятностью A	Ошибка

Пример.

TRANSFER .333,BL1,BL2

...

BL1 SEIZE PR1

...

BL2 QUEUE QPR1

В данном примере с вероятностью 0.667 транзакт переходит в блок BL1, а с вероятностью 0.333 – в блок BL2.

1.10. Пример

Каждые 10 ± 6 минут на обработку на станке поступает деталь. Время обработки детали на станке равно 9 ± 2 минуты. В каждый момент времени на станке может обрабатываться только одна деталь. Детали, ожидающие обработки, временно хранятся на стеллаже около станка. После обработки детали выполняется выходной контроль, в результате которого в среднем 10% деталей бракуются. Требуется выполнить моделирование изготовления на станке 1000 исправных деталей.

Модель системы, описанной в примере, представляет собой систему массового обслуживания, показанную на рис. 1.




Рис. 1. Структура модели для примера.

Модель состоит из четырех основных элементов идущих на рисунке слева направо:

1) ввод транзактов (деталей) в модель. Показан на рисунке стрелкой в левой части модели с надписью “ 10 ± 6 мин.”. Это означает, что с интервалом 10 ± 6 мин. в модель поступают транзакты.

2) очередь (стеллаж). Показана на рисунке прямоугольником с вертикальными линиями. Транзакты проходят свободно через очередь, если расположенное за ней устройство свободно. Если же устройство занято, то транзакт задерживается в очереди до тех пор, пока устройство не освободится.

3) устройство (станок). Показано на рисунке прямоугольником. После того, как в устройство входит транзакт, устройство в течение некоторого времени (9 ± 2 мин.) считается занятым (станок обрабатывает деталь). По истечении данного времени транзакт выходит из устройства, и оно может начать обслуживание другого транзакта.

4) вывод транзактов из модели. Показан на рисунке стрелкой в правой части рисунка. Здесь выполняется решение о том, является ли деталь бракованной, и вывод транзактов из модели.

Соответствующая программа на языке GPSS имеет вид:

```

GENERATE 10,6 ; Ввод в модель транзактов с интервалом 10+-6
QUEUE Stellaj ; Вход транзакта в очередь
SEIZE Stanok ; Вход транзакта в станок
DEPART Stellaj ; Выход транзакта из очереди
ADVANCE 9,2 ; Задержка транзакта на обработку 9+-2
RELEASE Stanok ; Выход транзакта из станка
TRANSFER .100,GOOD,BAD ; Выходной контроль
GOOD TERMINATE 1 ; Вывод заявки из модели
BAD TERMINATE

```

Здесь присутствуют следующие блоки языка GPSS:

GENERATE 10, 6 – данный блок моделирует ввод транзактов в модель, что соответствует поступлению деталей на обработку;

QUEUE Stellaj, DEPART Stellaj – данные блоки моделируют вход транзакта в очередь и выход транзакта из очереди (где Stellaj – имя очереди), что соответствует размещению детали на стеллаже;

SEIZE Stanok, RELEASE Stanok – данные блоки моделируют вход транзакта в устройство и выход транзакта из устройства, что соответствует началу и концу обработки детали;

ADVANCE 9, 2 – данный блок моделирует задержку транзакта в устройстве на время обработки детали;

TRANSFER 0.100, GOOD, BAD – данный блок с вероятностью 0,1 направляет транзакты на блок с меткой BAD, а с вероятностью 0,9 – на блок с меткой GOOD;

TERMINATE 1 – вывод транзакта из модели с уменьшением счетчика завершения. Соответствует производству исправной детали.

TERMINATE – вывод транзакта из модели без уменьшения счетчика завершения. Соответствует производству бракованной детали.

2. Порядок выполнения работы

1. Изучите теоретический материал.
2. Разработайте модель САПР в соответствии с вашим вариантом. Номер варианта соответствует номеру компьютера в лаборатории.
3. Выполните моделирование и определите среднее время ожидания рабочей станцией ответа от сервера.
4. Сформулируйте выводы по работе и ответьте на контрольные вопросы.

3. Содержание отчета

1. Название работы.
2. Цель работы.
3. Задание на лабораторную работу.
4. Описание основных блоков языка GPSS.
5. Программа на языке GPSS.
6. Результаты моделирования.
7. Выводы по работе.

4. Задание на лабораторную работу

В составе САПР используются рабочая станция и сервер, объединенные общей локальной сетью. Каждые T миллисекунд ра-

бочая станция посылает запрос к серверу. Запрос передается по сети (в каждый момент времени данные по сети может передавать только один компьютер), обрабатывается сервером, а затем по сети на рабочую станцию передается ответ от сервера. Время передачи сообщения по сети равно 20 мс. Данные, ожидающие отправки по сети, хранятся во внутренней памяти компьютера. Сервер обрабатывает запросы в многопоточном режиме. В каждый момент времени может обрабатываться не более K запросов. Запросы, ожидающие обслуживания на сервере, хранятся в его оперативной памяти. Время обработки сервером одного запроса равно t .

Требуется определить среднее время ожидания рабочей станцией ответа от сервера при обработке 10000 запросов.

Исходные данные по вариантам приведены в таблице:

Номер варианта	K	$T, \text{мс}$	$t, \text{мс}$
1	2	100±15	170±50
2	3	80±10	230±10
3	2	300±100	400±150
4	3	350±150	800±200
5	2	150±30	270±30
6	3	100±15	290±20
7	2	80±10	150±25
8	3	200±100	500±80
9	2	500±150	850±150
10	3	150±30	400±70

5. Контрольные вопросы

1. Формат операторов языка GPSS.
2. Назначение и параметры блока GENERATE.
3. Назначение и параметры блока TERMINATE.
4. Назначение и параметры блока ADVANCE.
5. Назначение и параметры блоков QUEUE, DEPART.
6. Назначение и параметры блоков SEIZE, RELEASE.
7. Назначение и параметры блоков ENTER, LEAVE.
8. Назначение и параметры блока TRANSFER.

Библиографический список

1. Томашевский В., Жданова Е. Имитационное моделирование в среде GPSS. – М.: Бестселлер, 2003. – 416 с.

2. Рыжиков Ю.И. Имитационное моделирование. Теория и технологии. – СПб.: КОРОНА принт; М.: Альтекс-А, 2004. – 384 с., ил.